

F07TWF (CTRTRI/ZTRTRI) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

F07TWF (CTRTRI/ZTRTRI) computes the inverse of a complex triangular matrix.

2 Specification

```
SUBROUTINE F07TWF(UPLO, DIAG, N, A, LDA, INFO)
ENTRY      ctrtri(UPLO, DIAG, N, A, LDA, INFO)
INTEGER    N, LDA, INFO
complex  A(LDA,*)
CHARACTER*1 UPLO, DIAG
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine forms the inverse of a complex triangular matrix A . Note that the inverse of an upper (lower) triangular matrix is also upper (lower) triangular.

4 References

- [1] Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

5 Parameters

- 1:** UPLO — CHARACTER*1 *Input*
On entry: indicates whether A is upper or lower triangular as follows:
- if UPLO = 'U', then A is upper triangular;
 - if UPLO = 'L', then A is lower triangular.
- Constraint:* UPLO = 'U' or 'L'.
- 2:** DIAG — CHARACTER*1 *Input*
On entry: indicates whether A is a non-unit or unit triangular matrix as follows:
- if DIAG = 'N', then A is a non-unit triangular matrix;
 - if DIAG = 'U', then A is a unit triangular matrix; the diagonal elements are not referenced and are assumed to be 1.
- Constraint:* DIAG = 'N' or 'U'.
- 3:** N — INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.

4: A(LDA,*) — *complex* array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1,N)$.

On entry: the n by n triangular matrix A . If $\text{UPLO} = \text{'U'}$, A is upper triangular and the elements of the array below the diagonal are not referenced; if $\text{UPLO} = \text{'L'}$, A is lower triangular and the elements of the array above the diagonal are not referenced. If $\text{DIAG} = \text{'U'}$, the diagonal elements of A are not referenced, but are assumed to be 1.

On exit: A is overwritten by A^{-1} , using the same storage format as described above.

5: LDA — INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F07TWF (CTRTRI/ZTRTRI) is called.

Constraint: $\text{LDA} \geq \max(1,N)$.

6: INFO — INTEGER *Output*

On exit: $\text{INFO} = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

$\text{INFO} < 0$

If $\text{INFO} = -i$, the i th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

$\text{INFO} > 0$

If $\text{INFO} = i$, a_{ii} is zero and the matrix A is singular.

7 Accuracy

The computed inverse X satisfies

$$|XA - I| \leq c(n)\epsilon|X||A|,$$

where $c(n)$ is a modest linear function of n , and ϵ is the *machine precision*.

Note that a similar bound for $|AX - I|$ cannot be guaranteed, although it is almost always satisfied.

The computed inverse satisfies the forward error bound

$$|X - A^{-1}| \leq c(n)\epsilon|A^{-1}||A||X|.$$

See Du Croz and Higham [1].

8 Further Comments

The total number of real floating-point operations is approximately $\frac{4}{3}n^3$.

The real analogue of this routine is F07TJF (STRTRI/DTRTRI).

9 Example

To compute the inverse of the matrix A , where

$$A = \begin{pmatrix} 4.78 + 4.56i & 0.00 + 0.00i & 0.00 + 0.00i & 0.00 + 0.00i \\ 2.00 - 0.30i & -4.11 + 1.25i & 0.00 + 0.00i & 0.00 + 0.00i \\ 2.89 - 1.34i & 2.36 - 4.25i & 4.15 + 0.80i & 0.00 + 0.00i \\ -1.89 + 1.15i & 0.04 - 3.69i & -0.02 + 0.46i & 0.33 - 0.26i \end{pmatrix}.$$

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      F07TWF Example Program Text
*      Mark 16 Release. NAG Copyright 1993.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
INTEGER          NMAX, LDA
PARAMETER       (NMAX=8,LDA=NMAX)
CHARACTER       DIAG
PARAMETER       (DIAG='N')
*      .. Local Scalars ..
INTEGER          I, IFAIL, INFO, J, N
CHARACTER       UPLO
*      .. Local Arrays ..
complex        A(LDA,NMAX)
CHARACTER       CLABS(1), RLABS(1)
*      .. External Subroutines ..
EXTERNAL        X04DBF, ctrtri
*      .. Executable Statements ..
WRITE (NOUT,*) 'F07TWF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN

*
*      Read A from data file
*
      READ (NIN,*) UPLO
      IF (UPLO.EQ.'U') THEN
        READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
      ELSE IF (UPLO.EQ.'L') THEN
        READ (NIN,*) ((A(I,J),J=1,I),I=1,N)
      END IF

*
*      Compute inverse of A
*
      CALL ctrtri(UPLO,DIAG,N,A,LDA,INFO)

*
*      Print inverse
*
      WRITE (NOUT,*)
      IFAIL = 0
      CALL X04DBF(UPLO,DIAG,N,N,A,LDA,'Bracketed','F7.4','Inverse',
+             'Integer',RLABS,'Integer',CLABS,80,0,IFAIL)
      END IF
      STOP
*
      END

```

9.2 Program Data

F07TWF Example Program Data

```

4                                     :Value of N
'L'                                   :Value of UPLO
( 4.78, 4.56)
( 2.00,-0.30) (-4.11, 1.25)
( 2.89,-1.34) ( 2.36,-4.25) ( 4.15, 0.80)
(-1.89, 1.15) ( 0.04,-3.69) (-0.02, 0.46) ( 0.33,-0.26) :End of matrix A

```

9.3 Program Results

F07TWF Example Program Results

Inverse

```

                                     1             2             3             4
1 ( 0.1095,-0.1045)
2 ( 0.0582,-0.0411) (-0.2227,-0.0677)
3 ( 0.0032, 0.1905) ( 0.1538,-0.2192) ( 0.2323,-0.0448)
4 ( 0.7602, 0.2814) ( 1.6184,-1.4346) ( 0.1289,-0.2250) ( 1.8697, 1.4731)

```
